Supplement 3. Mplus syntax for multiple-group confirmatory factor analyses


TITLE:	Configural invariance model, theta parameterisation. Freely varying factor loadings and response thresholds. Factor means fixed to zero and residual variances fixed to one across groups for model identification. Item psy2 = thought interference, item psy3 = paranoia, item psy4 = strange experience, item psy5 = hallucination.

DATA: 	FILE IS C:\...

VARIABLE: 	NAMES ARE caseid pweight psy2 psy3 psy4 psy5 ethnic3;
	USEVARIABLES ARE psy2 psy3 psy4 psy5;
	CATEGORICAL ARE psy2 psy3 psy4 psy5;
	IDVARIABLE IS caseid;
	WEIGHT IS pweight;
	GROUPING IS ethnic3 (1=WB, 2=WI, 3=AC, 4=IND, 5=PAK);

MODEL: 	psyc BY psy2 psy3 psy4 psy5;

ANALYSIS:	PARAMETERIZATION IS theta;

MODEL WI:	PSYC BY psy3 psy4 psy5; ! factor loadings free, except for marker item 
[psy2$1 psy3$1 psy4$1 psy5$1]; ! response thresholds free	
	[psyc@0]; ! factor mean fixed to zero
	psy2@1 psy3@1 psy4@1 psy5@1; ! residual variances fixed to one

MODEL AC:	PSYC BY psy3 psy4 psy5; 
[psy2$1 psy3$1 psy4$1 psy5$1];	
	[psyc@0]; 
	psy2@1 psy3@1 psy4@1 psy5@1; 

MODEL IND:	PSYC BY psy3 psy4 psy5; 
[psy2$1 psy3$1 psy4$1 psy5$1];	
	[psyc@0]; 
	psy2@1 psy3@1 psy4@1 psy5@1; 

MODEL PAK:	PSYC BY psy3 psy4 psy5; 
[psy2$1 psy3$1 psy4$1 psy5$1]; 
	[psyc@0]; 
	psy2@1 psy3@1 psy4@1 psy5@1; 

OUTPUT:	STANDARDIZED;
	RESIDUAL;
	CINTERVAL;
	TECH1;

PLOT:	TYPE IS plot2;

SAVEDATA:	DIFFTEST IS "configural invariance_difftest.dat"

Supplement 3. Continued


TITLE:	Strong factorial invariance model, theta parameterisation. Equality constraints on factor loadings and response thresholds across groups. Factor mean fixed to zero among British Whites, but estimated freely in other groups. Residual variances fixed to one among British Whites, but estimated freely in other groups. Item psy2 = thought interference, item psy3 = paranoid symptoms, item psy4 = strange experience, item psy5 = hallucination. 

DATA: 	FILE IS C:\...

VARIABLE: 	NAMES ARE caseid pweight psy2 psy3 psy4 psy5 ethnic3;
	USEVARIABLES ARE psy2 psy3 psy4 psy5;
	CATEGORICAL ARE psy2 psy3 psy4 psy5;
	IDVARIABLE IS caseid;
	WEIGHT IS pweight;
	GROUPING IS ethnic3 (1=WB, 2=WI, 3=AC, 4=IND, 5=PAK);

MODEL: 	psyc BY psy2 psy3 psy4 psy5;

ANALYSIS:	PARAMETERIZATION IS theta;
	DIFFTEST IS "configural invariance_difftest.dat";

OUTPUT:	STANDARDIZED;
	RESIDUAL;
	CINTERVAL;
	TECH1;
	MODINDICES(3.84);

PLOT:	TYPE IS plot2;

SAVEDATA:	DIFFTEST IS "strong factorial invariance_difftest.dat"











Supplement 3. Continued


TITLE:	Strict factorial invariance model, theta parameterisation. Equality constraints on factor loadings and response thresholds across groups. Factor mean fixed to zero among British Whites, but estimated freely in other groups. All residual variances fixed to one across groups. Item psy2 = thought interference, item psy3 = paranoid symptoms, item psy4 = strange experience, item psy5 = hallucination. 

DATA: 	FILE IS C:\...

VARIABLE: 	NAMES ARE caseid pweight psy2 psy3 psy4 psy5 ethnic3;
	USEVARIABLES ARE psy2 psy3 psy4 psy5;
	CATEGORICAL ARE psy2 psy3 psy4 psy5;
	IDVARIABLE IS caseid;
	WEIGHT IS pweight;
	GROUPING IS ethnic3 (1=WB, 2=WI, 3=AC, 4=IND, 5=PAK);

MODEL: 	psyc BY psy2 psy3 psy4 psy5; 

MODEL AC: 	psy2@1 psy3@1 psy4@1 psy5@1; 

MODEL IND: 	psy2@1 psy3@1 psy4@1 psy5@1; 

MODEL PAK: 	psy2@1 psy3@1 psy4@1 psy5@1;

ANALYSIS:	PARAMETERIZATION IS theta;
	DIFFTEST IS "strong factorial invariance_difftest.dat"

OUTPUT:	STANDARDIZED;
	RESIDUAL;
	CINTERVAL;
	TECH1;
	MODINDICES(3.84);

PLOT:	TYPE IS plot2;








Supplement 3. continued


TITLE:	Partial strict factorial invariance model, theta parameterisation. Equality constraints on factor loadings and item thresholds across groups. Factor mean fixed to zero among British Whites, but estimated freely in other groups. Residual variances for thought interference, strange experience and hallucination fixed to one across groups. Residual variance for paranoid symptoms estimated freely across groups. Item psy2 = thought interference, item psy3 = paranoid symptoms, item psy4 = strange experience, item psy5 = hallucination.  

DATA: 	FILE IS C:\...

VARIABLE: 	NAMES ARE caseid pweight psy2 psy3 psy4 psy5 ethnic3;
	USEVARIABLES ARE psy2 psy3 psy4 psy5;
	CATEGORICAL ARE psy2 psy3 psy4 psy5;
	IDVARIABLE IS caseid;
	WEIGHT IS pweight;
	GROUPING IS ethnic3 (1=WB, 2=WI, 3=AC, 4=IND, 5=PAK);

MODEL: 	psyc BY psy2 psy3 psy4 psy5; 

MODEL AC: 	psy2@1 psy4@1 psy5@1; 

MODEL IND: 	psy2@1 psy4@1 psy5@1;

MODEL PAK: 	psy2@1 psy4@1 psy5@1; 

ANALYSIS:	PARAMETERIZATION IS theta;
	DIFFTEST IS "strong factorial invariance_difftest.dat"

OUTPUT:	STANDARDIZED;
	RESIDUAL;
	CINTERVAL;
	TECH1;
	MODINDICES(3.84);

PLOT:	TYPE IS plot2;

